

Hans Walser, [20071007a]

Fibonacci und reguläre Vielecke

1 Worum es geht

Für $m \in \mathbb{N}$, $m \geq 2$ führt die verallgemeinerte Fibonacci-Rekursion

$$a_{n+2} = \left(1 + e^{i\frac{2\pi}{m}}\right)a_{n+1} - e^{i\frac{2\pi}{m}}a_n$$

Für beliebige Startwerte a_0, a_1 zu einem regulären m -Eck.

2 Beispiel

Für $m = 7$ und $a_0 = 3 + 4i$ und $a_1 = 1 + 2i$ erhalten wir mit dem MuPAD-Programm

```
m:=7:

p:=float(1/2*(1+exp(I*2*PI/m))):
q:=float(-exp(I*2*PI/m)):

a[0]:=3+4*I:
a[1]:=1+2*I:

for n from 0 to m do
  a[n+2]:=2*p*a[n+1]+q*a[n];
end_for:

Punkt:=n->plot::Point2d([Re(a[n]), Im(a[n])], PointSize=2,
PointColor=[1,0,0]):


Startpunkt0:=plot::Point2d([Re(a[0]), Im(a[0])],
PointSize=3, PointColor=[0,1/2,0]):
Startpunkt1:=plot::Point2d([Re(a[1]), Im(a[1])],
PointSize=3, PointColor=[0,1/2,0]):

Polygon:=n->plot::Polygon2d([[Re(a[n]), Im(a[n])],
[Re(a[n+1]), Im(a[n+1])]],
LineWidth=0.5, LineColor=[0,0,1]):

Ursprung:=plot::Point2d([0, 0], PointSize=2,
PointColor=[0,0,0]):

plot( Polygon(n)$n=0..m, Punkt(n)$n=0..m, Startpunkt0,
Startpunkt1, Ursprung, Scaling=Constrained, TicksDistance=1,
TicksBetween=0,
AxesLineWidth=0.5, AxesLineColor=[0,0,0], AxesTitle-
Font=["Times", 12, Italic], TicksLabelFont=["Times", 12],
Width=80, Height=80);
```

in der komplexen Ebene:



$$m = 7$$

3 Beweise

3.1 Umformen

Die Rekursion

$$a_{n+2} = \left(1 + e^{i\frac{2\pi}{m}}\right)a_{n+1} - e^{i\frac{2\pi}{m}}a_n$$

lässt sich umformen:

$$a_{n+2} = \left(1 + e^{i\frac{2\pi}{m}}\right)a_{n+1} - e^{i\frac{2\pi}{m}}a_n = a_{n+1} + (a_{n+1} - a_n)e^{i\frac{2\pi}{m}}$$

Im Punkt a_{n+1} wird also die um den Winkel $\frac{2\pi}{m}$ verdrehte Vorgängerstrecke von a_n nach a_{n+1} angehängt. Dies führt zu einem gleichseitigen Polygonzug mit jeweiliger Richtungsänderung $\frac{2\pi}{m}$, für $m \in \mathbb{N}$, $m \geq 2$ also zu einem regulären m -Eck.

3.2 Formel von Binet

Allgemein gilt: Für eine Fibonacci-Folge mit der Rekursion

$$a_{n+2} = 2pa_{n+1} + qa_n, \quad p, q \in \mathbb{C}$$

und Startwerten a_0 und a_1 gilt mit den Hilfsgrößen

$$\gamma_1 = p + (p^2 + q)^{\frac{1}{2}} \quad \text{und} \quad \gamma_2 = p - (p^2 + q)^{\frac{1}{2}}$$

die explizite Formel von Binet:

$$a_n = \frac{1}{\gamma_1 - \gamma_2} \left((a_1 - a_0\gamma_2)\gamma_1^n + (a_0\gamma_1 - a_1)\gamma_2^n \right)$$

Dies kann induktiv bewiesen werden.

In unserem Fall ist:

$$p = \frac{1}{2} \left(1 + e^{i\frac{2\pi}{m}} \right)$$

$$q = -e^{i\frac{2\pi}{m}}$$

Diese Schreibweise wurde bereits im Programm zu obigem Beispiel verwendet. Wir erhalten:

$$\begin{aligned} \gamma_1 &= p + (p^2 + q)^{\frac{1}{2}} = \frac{1}{2} \left(1 + e^{i\frac{2\pi}{m}} \right) + \left(\frac{1}{4} \left(1 + e^{i\frac{2\pi}{m}} \right)^2 - e^{i\frac{2\pi}{m}} \right)^{\frac{1}{2}} \\ &= \frac{1}{2} \left(1 + e^{i\frac{2\pi}{m}} \right) + \frac{1}{2} \left(1 - e^{i\frac{2\pi}{m}} \right) = 1 \\ \gamma_2 &= p - (p^2 + q)^{\frac{1}{2}} = \frac{1}{2} \left(1 + e^{i\frac{2\pi}{m}} \right) - \left(\frac{1}{4} \left(1 + e^{i\frac{2\pi}{m}} \right)^2 - e^{i\frac{2\pi}{m}} \right)^{\frac{1}{2}} \\ &= \frac{1}{2} \left(1 + e^{i\frac{2\pi}{m}} \right) - \frac{1}{2} \left(1 - e^{i\frac{2\pi}{m}} \right) = e^{i\frac{2\pi}{m}} \end{aligned}$$

und damit die Formel von Binet:

$$a_n = \frac{1}{1 - e^{i\frac{2\pi}{m}}} \left(\left(a_1 - a_0 e^{i\frac{2\pi}{m}} \right) + (a_0 - a_1) e^{i\frac{2\pi}{m}n} \right) = \frac{a_1 - a_0 e^{i\frac{2\pi}{m}}}{1 - e^{i\frac{2\pi}{m}}} + \frac{a_0 - a_1}{1 - e^{i\frac{2\pi}{m}}} e^{i\frac{2\pi}{m}n}$$

Dies führt zu einem regulären m -Eck mit dem Mittelpunkt

$$\frac{a_1 - a_0 e^{i\frac{2\pi}{m}}}{1 - e^{i\frac{2\pi}{m}}}$$

und dem Umkreisradius:

$$\frac{a_0 - a_1}{1 - e^{i\frac{2\pi}{m}}}$$

Kontrolle an obigem Beispiel: Das erweiterte Programm

```
m:=7:
```

```
p:=float(1/2*(1+exp(I*2*PI/m))):
```

```
q:=float(-exp(I*2*PI/m)):
```

```
a[0]:=3+4*I:
```

```
a[1]:=1+2*I:
```

```
for n from 0 to m do
```

```
  a[n+2]:=2*p*a[n+1]+q*a[n];
```

```
end_for:
```

```
Punkt:=n->plot::Point2d([Re(a[n]), Im(a[n])], PointSize=2,  
PointColor=[1,0,0]):
```

```
Startpunkt0:=plot::Point2d([Re(a[0]), Im(a[0])],
```

```

PointSize=3, PointColor=[0,1/2,0]):
Startpunkt1:=plot::Point2d([Re(a[1]), Im(a[1])],
PointSize=3, PointColor=[0,1/2,0]):

Polygon:=n->plot::Polygon2d([[Re(a[n]), Im(a[n])],
[Re(a[n+1]), Im(a[n+1])]],
LineWidth=0.5, LineColor=[0,0,1]):

Ursprung:=plot::Point2d([0, 0], PointSize=2,
PointColor=[0,0,0]):

Mitte:=float((a[1]-a[0]*exp(I*2*PI/m))/(1-exp(I*2*PI/m))):

xM:=Re(Mitte):
yM:=Im(Mitte):

r:=abs(float((a[0]-a[1])/(1-exp(I*2*PI/m)))):


Mittelpunkt:=plot::Point2d([xM, yM], PointSize=2,
PointColor=[1,0,1]):

Umkreis:=plot::Curve2d([xM+r*cos(t), yM+r*sin(t)],
t=0..2*PI, LineColor=[1,0,1]):

plot( Umkreis, Polygon(n)$n=0..m, Punkt(n)$n=0..m,
Startpunkt0, Startpunkt1, Ursprung, Mittelpunkt,
Scaling=Constrained, TicksDistance=1, TicksBetween=0,
AxesLineWidth=0.5, AxesLineColor=[0,0,0],
AxesTitleFont=["Times", 12, Italic],
TicksLabelFont=["Times", 12],
Width=80, Height=80);

```

liefert:



Mit Mittelpunkt und Umkreis


4 Sternfiguren

Für $m \in \mathbb{N}, m \geq 2$ und $k \in \mathbb{N}, k \geq 2, k < m$ führt die verallgemeinerte Fibonacci-Rekursion

$$a_{n+2} = \left(1 + e^{i\frac{2k\pi}{m}}\right)a_{n+1} - e^{i\frac{2k\pi}{m}}a_n$$

Für beliebige Startwerte a_0, a_1 zu einem regulären $\left\{\frac{m}{k}\right\}$ -Eck.

Beispiel für $m = 7$ und $k = 3$:



Sternfigur